



北京精电蓬远显示技术有限公司

VARITRONIX PENGYUAN LTD.

内藏 **HD61830** 控制驱动器
图形液晶显示模块

使用手册

(第一版)

2001 年 7 月

 精电国际集团成员

目 录

注意事项	2
第一章	前言.....	4
第二章	HD61830的一般介绍.....	4
第三章	内藏HD61830的液晶模块.....	10
第四章	内藏HD61830型液晶模块的应用.....	11
附录一	HD61830内藏字符集.....	20
附录二	精电蓬远公司配套产品.....	21
附录三	精电蓬远公司维修服务规范.....	21
附录四	内藏HD61830控制驱动器图形液晶显示模块目录 及其尺寸图.....	22

注意事项



十分感谢您购买我公司的产品，在使用前请您首先仔细阅读以下注意事项，以免给您造成不必要的损失，您在使用过程中遇到困难时，请拨打我们的技术服务电话，我们将尽力为您提供服务和帮助。

一、处理保护膜

在装好的模块成品表面贴有一层保护膜，以防在装配时沾污显示表面，在整机装配结束前不得揭去，以免弄脏或损坏显示面。



二、加装衬垫

在模块与前面板之间最好加装一块约0.1毫米左右的衬垫。面板还应保持平整，以免在装配后产生扭曲，并可提高其抗振性能。

三、严防静电

模块中的控制、驱动电路是低压、低功耗的CMOS电路，极易被静电击穿，静电击穿是一种不可修复的损坏，而人体有时会产生高达几十伏或上百伏的高压静电，所以，在操作、装配以及使用中都应极其小心，严防静电。为此：

1. 不要用手随意去摸外引线、电路板上的电路及金属框；
2. 如必须直接接触时，应使人体与模块保持在同一电位，或将人体良好接地；
3. 焊接使用的烙铁和操作用的电动工具必须良好接地，没有漏电；
4. 不得使用真空吸尘器进行清洁处理，因为它会产生很强的静电；
5. 空气干燥，也会产生静电，因此，工作间湿度应在RH60%以上；
6. 取出或放回包装袋或移动位置时，也需小心，防止产生静电。不要随意更换包装或舍弃原包装。



四、装配操作时的注意事项


1. 模块是经过精心设计组装而成的，请勿随意自行加工、修整；
2. 金属框爪不得随意扭动、拆卸；
3. 不要随意修改加工PCB板外形、装配孔、线路极其部件；
4. 不得修改导电胶条；
5. 不得修改任何内部支架；
6. 不要碰、摔、折曲、扭动模块。



五、焊接

在焊接模块外引线、接口电路时，应按如下规程进行操作。

1. 烙铁头温度小于280℃；
2. 焊接时间小于3~4s；
3. 焊接材料：共晶型、低熔点；
4. 不要使用酸性助焊剂；
5. 重复焊接不要超过3次，且每次重复需间隔5分钟。




六、模块的使用与保养

1. 模块的外引线决不允许接错，在您想调试液晶模块时，请注意正确接线，尤其是正、负电源的接线不能有错，否则可能造成过流、过压、烧毁电路上的芯片等对液晶模块元器件有损的现象；
2. 模块在使用时，接入电源及断开电源，必须在正电源稳定接入以后，才能输入信号电平。如在电源稳定前或断开后输入信号电平，有可能损坏模块中的IC及电路；
3. 模块使用时，接入逻辑电源和驱动电源的顺序应是先逻辑电源，后驱动电源；断电时，应先驱动电源，后逻辑电源。这样做有助于保持屏的良好显示效果和避免在上电、断电时的电压冲击损坏。所以推荐使用带控制的DC-DC电源做为模块的驱动电源。
4. 点阵液晶模块显示时的对比度、视角与温度、驱动电压关系很大，所以，如果VEE调整过高，不仅会影响显示，还会缩短模块的使用寿命；
5. 因为液晶材料的物理特性，液晶的对比度会随着温度的变化而相应变化，所以，您加的负电压值应该随温度作相应的调整，大致是温度变化 10°C ，电压变化1伏。为满足这一要求，您可做一个温度补偿电路，或者安排一个电位器，随温度调整负电压值；
6. 不应在规定工作温度范围以外使用，并且不应在超过存储极限温度的范围外存储，如果温度低于结晶温度，液晶就会结晶，如果温度过高，液晶将变成各向同性的液体，破坏分子取向，使器件报废；
7. 用力按压显示部分，会产生异常显示。这时切断电源，稍待片刻，重新上电，即恢复正常；
8. 液晶显示器件或模块表面结雾时，不要通电工作，因为这将引起电极化学反应，产生断线；
9. 长期用于阳光及强光下时，被遮部分会产生残留影像。

七、模块的存储

若长期（如几年以上）存储，我们推荐以下方式：

1. 装入聚乙稀口袋（最好有防静电涂层）并将口封住；
2. 在 -10°C ~ $+35^{\circ}\text{C}$ 之间存储；
3. 放暗处，避强光；
4. 决不能在表面压放任何物品；
5. 严格避免在极限温/湿度条件下存放。



八、责任范围及维修

在您购买液晶显示模块时，精电蓬远将会为您做显示模块的检测，确保您所买的显示模块为完好的器件；在您使用过程中因不小心将显示模块损坏，您可送至精电蓬远维修部修理。液晶模块如果出现屏被损坏的问题，比如玻璃面破损、玻璃屏角碎裂等等，将无法进行修理，您的液晶模块只能报废。

公司产品质量都是通过国际质量认证及时间的考验的，请用户放心使用。如果在使用过程中发现问题，可以拿到我公司质量检测部检测，确实是质量问题，而且购买时间在一年之内，我公司给予换货处理。因为用户使用不当（例如静电、焊接、连线不当、过流、过压使用等）所造成的损坏，可送到我们维修部修理（具体事宜请参照北京精电蓬远显示技术有限公司维修服务规范）。

内藏HD61830控制器的液晶显示模块的驱动电压 (仅供参考)

测试温度: 20 °C	电源电压: 4.9±0.1 v		
MGLS-8032B	- 4.0 V	MGLS-32064	-10.0V
MGLS-8464	- 5.2 V	MGLS-32064-HT	-10.0V
MGLS-240128	- 15.0 V		
MGLS-240128-HT	- 15.0 V		
MGLS-16080	单电源供电, 需外接电位器		

第一章 前言

HD61830是一种图形液晶显示控制器, 它可与8位微处理器接口, 管理64K显示RAM, 内部时序发生器产生点阵液晶显示驱动信号。本手册将对HD61830作一般介绍, 然后详细叙述内藏HD61830控制器的液晶显示模块的应用。


第二章 HD61830的一般介绍

一、HD61830的特点

1. HD61830是图形液晶显示控制器, 可直接与M6800系列时序的MPU接口;
2. 具有专用指令集, 可完成文本显示或图形显示的功能设置, 以及实现画面卷动、光标、闪烁、位操作等功能;
3. HD61830可管理64K显示RAM, 其中图形方式为64K, 字符方式为4K;
4. 内部字符发生器CGROM共有192种字符, 其中5×7字体160种, 5×11字体有32种, HD61830还可外接字符发生器, 使字符量达到256种;
5. HD61830可以静态方式显示至1/128占空比的动态方式显示。

二、HD61830的引脚说明及其功能


HD61830封装为60个引脚, 管脚图和功能原理图分别如下。



管脚功能如下：

符号	状态	名称	功能
DB0—DB7	三态	与MPU接口的数据总线	
/CS	输入	片选信号	低电平有效
R/W	输入	读、写选择信号	当R/W=1时MPU从HD61830读出数据，当R/W=0时MPU向HD61830写数据
RS	输入	寄存器选择	RS=1选通指令寄存器，RS=0选通数据寄存器
E	输入	使能信号	在E下降沿写数据，E为高电平时为读数据
CR,R,C		RC振荡器引出端	接入振荡电阻和电容即可起振，使HD61830工作
RES	输入	复位信号	低电平有效
MA0—MA15	输出	外接显示RAM的地址输出端	文本方式下，MA12—MA15为外接字符库CGROM的地址输出端，所以HD61830只能管理4K的文本显示RAM
MD0—MD7	三态	显示RAM数据总线	
RD0—RD7	输入	外接CGROM数据输入端	
WE	输出	显示RAM的写信号	
CL1,CL2	输出	LCD驱动器的显示数据的锁存信号和移动时钟	
FLM	输出	显示同步器帧信号	
MA,MB	输出	液晶交流驱动信号	
D1,D2	输出	显示数据串行输出	液晶屏为单屏结构时，只使用D1；当液晶屏为双屏结构时，D1、D2分别用于上、下半屏
CP0	输出	时钟信号输出端	
SYNC	三态	多片HD61830并联使用时的同步信号	

功能原理图:




三、与MPU接口的读、写操作时序:

HD61830 MPU Interface ($V_{CC} = 5V \pm 10\%$, $GND = 0V$, $T_a = -20$ to $+75^\circ C$)

Item	Symbol	Min	Typ	Max	Unit
Enable cycle time	t_{CYC}	1.0	—	—	μs
Enable pulse width	High level	t_{WEH}	0.45	—	μs
	Low level	t_{WEL}	0.45	—	μs
Enable rise time	t_{Er}	—	—	25	ns
Enable fall time	t_{Ef}	—	—	25	ns
Setup time	t_{AS}	140	—	—	ns
Data setup time	t_{DSW}	225	—	—	ns
Data delay time	t_{DDR}	—	—	225	ns *
Data hold time	t_{DHW}	10	—	—	ns
Address hold time	t_{AH}	10	—	—	ns
Output data hold time	t_{DH}	20	—	—	ns

Note: * The following load circuit is connected for specification:



当/CS为低电平时，RS、R/W、E的各种组合所实现的功能如下表：

RS	R/W	E	功能
0	0	下降沿	写数据或指令参数
0	1	高电平	读数据
1	0	下降沿	写指令代码
1	1	高电平	读忙标志位

四、指令系统

HD61830有13条指令，指令是由一个指令代码和一个功能参数组成。指令代码好似参数寄存器的地址代码，而参数才是实质的功能值。MPU向HD61830指令寄存器写入指令代码来选择参数寄存器，再通过数据寄存器向参数寄存器写入参数值，以实现功能的设置。

HD61830向MPU提供一个忙（BF）标志位：

BF=1表示当前HD61830处于内部运行状态，不接受MPU的访问（读忙标志位除外）；

BF=0表示HD61830允许MPU的访问。

MPU在访问HD61830时都要判断BF是否为0。

MPU可在RS=1下从数据总线D7位上读出BF标志值。

下面逐条介绍这些指令的设置：

1. 方式控制，指令代码为00H

该指令参数定义了显示方式。其参数格式为：

0	0	D5	D4	D3	D2	D1	D0
---	---	----	----	----	----	----	----

D0：字符发生器选择，D0=0为CGROM，D0=1为EXCGROM；

D1：显示方式选择，D1=0为文本方式，D1=1为图形方式；

D3与D2组合实现如下功能：

D3	D2	功能
0	0	光标禁止
0	1	启用光标
1	0	光标禁止，字符闪烁
1	1	光标闪烁

D4: 工作方式选择, D4=0为从方式, D4=1为主方式;

D5: 显示状态选择, D5=0为禁止显示, D5=1为启用显示。

需要注意的是, 使用图形方式 (D1=1) 时, 只能使D0=D2=D3=0。一般情况下, HD61830采用主方式 (D4=1)。当有两片HD61830并联时, 则其中一片为主方式, 一片为从方式。

2. 字体设置, 指令代码01H

该指令设置文本方式下字符的点阵大小, 指令参数格式:

VP-1	0	HP-1
------	---	------

VP: 字符点阵行数, 取值范围1~16;

HP: 字符点阵列数, 图形方式表示一字节显示数据的有效位数, HP的取值范围为6, 7, 8。

3. 显示域设置, 指令代码02H

该指令的指令参数格式如下:

0	HN-1
---	------

HN为一行显示所占的字节数, 其取值范围为2~128内的偶数值, 由HN和HP可得显示屏有效显示点列数 $N=HN \times HP$ 。

4. 帧设置, 指令代码03H

该指令的指令参数格式如下:

0	NX-1
---	------

NX为显示时的帧扫描行数, 其倒数即为占空比。对单屏结构的显示模块, NX即为其有效显示行数; 对双屏显示结构的模块则2NX为其有效显示行数。

5. 光标位置设置, 指令代码04H

文本方式下的光标为一行 (8×1) 点阵显示, 该指令用来指明光标在字符位中的第几行, 指令参数格式如下:

0	0	0	0	CP-1
---	---	---	---	------

其中CP表示光标在字符体中的行位置, CP取值范围在1~VP之间。CP>VP时光标将被禁止。

6. SADL设置, 指令代码08H

该指令的指令参数格式如下:

起始地址低8位 SADL

7. SADH设置, 指令代码09H

该指令的指令参数格式如下:

起始地址高8位 SADH

以上两条指令设置了显示缓冲区起始地址，它们的指令参数分别是该地址的低位和高位字节。该地址对应显示屏上左上角显示位，显示缓冲区单元（即RAM单元）与显示屏上的显示位的一一对应关系如下表所示：

SAD	SAD+1	SAD+HN-1
SAD+HN	SAD+HN+1	SAD+2HN-1
...
...
...
SAD+MHN	SAD+MHN+1	SAD+(M+1)HN-1

显示屏上一个显示位长为HP点阵。

8. CACL设置，指令代码0AH

该指令的指令参数格式如下：

光标地址低8位 CACL

9. CACH设置，指令代码0BH

该指令的指令参数格式如下：

光标地址高8位 CACH

以上两条指令设置光标地址指针，它们的指令参数即是该光标地址指针的低位和高位字节。其作用一是用来指示当前要读、写显示缓冲区单元的地址；二是用在文本方式下，指出光标或闪烁字符在显示屏上的位置。

由于光标地址计数器是16位加1计数器，当第N位从1变到0时，会引起第N+1位自动加1，因此当设置低8位地址时，若最高位MSB从1变为0就会引起高8位地址的最低位LSB加1。因此设置时应先设置低8位，再设置高8位，而且即使只需修改地址低8位，其高位也要跟着重新设置一次，以确保地址指针设置的万无一失。

10. 数据写，指令代码0CH

该指令将随后写入数据寄存器的数据送入光标地址指针指向的显示RAM单元。光标地址指针将随着每次数据的写入而自动加一修改。该指令功能的终止将由下一条指令的输入来完成。

11. 数据读，指令代码0DH

该指令代码写入后，紧跟着一次“空读”操作后，则可以连续读出光标地址指针所指向单元的内容。光标地址指针将随着每次数据的读出而自动加一修改，该指令功能的终止将由下一条指令的输入来完成。

12. 位清零，指令代码0EH

13. 位置1，指令代码0FH

以上两条指令的功能是将光标地址指针所指向的显示RAM单元中的某一位清零或置1。指令执行一次，光标地址指针自动加一。指令参数格式为：

0	0	0	0	0	NB-1
---	---	---	---	---	------

其中NB为要清零或置1的位址，取值1~8，对应该字节的LSB~MSB位。

以上列出了HD61830的所有指令。由于HD61830的指令代码好似参数寄存器的地址代码，所以在写入一个指令代码后向数据口写入的多个数据都将修改该指令代码所指的参数寄存器内的内容。

第三章 内藏HD61830的液晶模块


内藏HD61830的液晶模块有四种型号：

MGLS-8032B、MGLS-8464、MGLS-240128、MGLS32064


它们除了所驱动的液晶屏尺寸不同外，其电路特性都是一样的。

模块的电原理图如下：

MGLS-240128S 电原理图



MGLS-32064 电原理图



接口端管脚功能

1	2	3	4	5	6	7~14	15	16	17	18
GND	VCC	V0	RS	R/W	E	DB0~DB7	CS	RST	LED+	LED-

注：LED+与LED-为背光的电源端

模块的电参数

名称	符号	MIN	TYP	MAX	单位	测试条件
电源电压	Vcc	4.5	5.0	5.5	V	
输入高电压(TTH)	VIH	2.2	-	Vcc	V	
输入低电压(TTL)	VIL	0	-	0.8	V	
输出高电压(TTL)	VOH	2.4	-	Vcc	V	IOH=0.6 mA
输出低电压(TTL)	VOL	0	-	0.4	V	IOL=1.6 mA
电源功率	Pw1	-	10	15	mW	CR波形频率 fosc=500MHz
时钟频率	fosc	400	500	600	KHz	Cf=15pF±5%,Rf=39KΩ


第四章 内藏HD61830型液晶模块的应用

在内藏控制器型的液晶显示模块上，已经完成了控制器与液晶显示驱动器和显示缓冲区的接口工作，留给用户的仅仅是与MPU的接口。因此只需了解HD61830的指令系统及与MPU接口的工作时序，而无需对液晶显示驱动器及其与HD61830的接口作太多了解就可使用内藏HD61830的液晶显示模块。

本节以精电蓬远公司提供的演示板为实用电路，说明内藏HD61830液晶显示模块的接口电路。

一、直接访问方式接口电路及驱动程序

精电蓬远公司提供的演示板接口电路如下图所示：



驱动子程序如下：

```

A15 = /CS ; A9 = R/W ; A8 = RS
DW_ADD EQU 8000H ; 写数据口地址
DR_ADD EQU 8200H ; 读数据口地址
CW_ADD EQU 8100H ; 写指令口地址
CR_ADD EQU 8300H ; 读状态口地址
COM EQU 30H ; 指令代码寄存器
    
```

DAT EQU 31H ; 数据寄存器

1. 读状态位子程序

```

PR0: MOV DPTR,#CR_ADD ; 设置读状态口地址
PR01: MOVX A,@DPTR ; 读状态
JB ACC.7,PR01 ; 判状态位为"0"否?
RET
    
```

2. 写指令代码子程序

```

PR1: LCALL PR0 ; 写指令入口
MOV DPTR,#CW_ADD ; 设置写指令口地址
MOV A,COM ; 取指令代码
MOVX @DPTR,A ; 写入指令代码
RET
    
```

3. 写指令参数和数据子程序

```

PR2: LCALL PR0 ; 写数据入口
MOV DPTR,#DW_ADD ; 设置写数据口地址
MOV A,DAT ; 取指令参数或显示数据
MOVX @DPTR,A ; 写入参数或数据
RET
    
```


4. 读显示数据子程序

```

PR3: LCALL PR0 ; 读数据入口
MOV DPTR,#DR_ADD ; 设置读数据口地址
MOVX A,@DPTR ; 读显示数据
MOV DAT,A ; 存数据
RET
    
```

二、间接控制方式接口电路及驱动程序

间接控制方式就是使用并行接口与内藏HD61830液晶显示模块接口。MPU通过控制并行接口的输出状态实现对HD61830的控制。接口电路如下图所示。由于并行接口是唯一对液晶显示模块接口，所以液晶显示模块的片选直接接地即可。



驱动子程序如下：

RS	EQU	P3.2	；通道选择信号
RW	EQU	P3.3	；读/写选择信号
E	EQU	P3.4	；使能信号

1. 读状态位子程序

```

PR0:   SETB    RS           ; RS=1
        SETB    RW           ; R/W=1
PR01:  MOV     P1,#0FFH     ; 置P1口为"1"
        SETB    E           ; E=1
        MOV     A,P1        ; 读状态
        CLR     E           ; E=0
        JB     ACC.7,PR01    ; 判状态位为"0"否?
        RET

```

2. 写指令代码子程序

```

PR1:   LCALL   PR0          ; 写指令入口
        CLR    RW           ; R/W=0
        MOV    A,COM        ; 取指令代码
        MOV    P1,A        ; 写入P1口
        SETB   E           ; E=1
        CLR    E           ; E=0
        RET

```

3. 写指令参数和数据子程序

```

PR2:   LCALL   PR0          ; 写数据入口
        CLR    RS           ; RS=0
        CLR    RW           ; R/W=0
        MOV    A,DAT        ; 取指令参数或显示数据
        MOV    P1,A        ; 写入数据
        SETB   E           ; E=1
        CLR    E           ; E=0
        RET

```

4. 读数据子程序

```

PR3:   LCALL   PR0          ; 写数据入口
        CLR    RS           ; RS=0
        MOV    P1,#0FFH     ; 置P1口为"1"
        SETB   E           ; E=1
        MOV    A,P1        ; 读数据
        CLR    E           ; E=0
        MOV    DAT,A       ; 存数据
        RET

```

三、应用程序

示例一 初始化

由于HD61830不能“图文并茂”，所以初始化要根据使用需要而设置。
初始化子程序(文本方式)

INT:	MOV	COM,#00H	; 方式设置 *
	LCALL	PR1	
	MOV	DAT,#3CH	; 开显示,主方式,文本方式
	LCALL	PR2	; 开光标闪烁
	MOV	COM,#01H	; 字符体设置 *
	LCALL	PR1	
	MOV	DAT,#77H	; 字符体为8×8点阵
	LCALL	PR2	
	MOV	COM,#02H	; 显示域宽度设置 *
	LCALL	PR1	
	MOV	DAT,#27H	; 一行占显示RAM 40个字节
	LCALL	PR2	
	MOV	COM,#03H	; 帧信号设置 *
	LCALL	PR1	
	MOV	DAT,#3FH	; 64行扫描行
	LCALL	PR2	
	MOV	COM,#04H	; 光标形状设置
	LCALL	PR1	
	MOV	DAT,#07H	; 光标为底线形式
	LCALL	PR2	
	MOV	COM,#08H	; 显示起始地址设置
	LCALL	PR1	
	MOV	DAT,#00H	; 低字节为 00H
	LCALL	PR2	
	MOV	COM,#09H	; 显示起始地址设置
	LCALL	PR1	
	MOV	DAT,#00H	; 高字节为 00H
	LCALL	PR2	
	RET		
初始化子程序(图形方式)			
INT:	MOV	COM,#00H	; 方式设置
	LCALL	PR1	
	MOV	DAT,#32H	; 开显示,主方式,图形方式
	LCALL	PR2	
	MOV	COM,#01H	; 字符体设置
	LCALL	PR1	
	MOV	DAT,#77H	; 字符体为8×8点阵
	LCALL	PR2	
	MOV	COM,#02H	; 显示域宽度设置 *
	LCALL	PR1	
	MOV	DAT,#27H	; 一行占显示RAM 40个字节
	LCALL	PR2	
	MOV	COM,#03H	; 帧信号设置 *
	LCALL	PR1	
	MOV	DAT,#3FH	; 64行扫描行
	LCALL	PR2	
	MOV	COM,#08H	; 显示起始地址设置
	LCALL	PR1	
	MOV	DAT,#00H	; 低字节为 00H

```

LCALL PR2
MOV COM,#09H ; 显示起始地址设置
LCALL PR1
MOV DAT,#00H ; 高字节为 00H
LCALL PR2
RET

```

注：* 标志的指令参数要根据所控制的液晶显示驱动系统的规定而设置

初始化示例程序

```

MAIN: MOV SP,#60H
      ANL P1,#0E3H ; RS=R/W=E=0
      LCALL INT
      LCALL CLEAR

```

示例二 清显示缓冲区子程序

```

CLEAR: MOV R2,#00H
      MOV R3,#10H
      MOV COM,#0AH
      LCALL PR1
      MOV DAT,#00H
      LCALL PR2
      MOV COM,#0BH
      LCALL PR1
      MOV DAT,#00H
      LCALL PR2
      MOV COM,#0CH
      LCALL PR1
      MOV DAT,#00H
CLR1:  LCALL PR2
      DJNZ R2,CLR1
      DJNZ R3,CLR1
      RET

```

示例三 西文字符写入

该示例提供了文本方式下的西文显示的操作，包括显示RAM地址的设置，西文字符显示，光标操作以及读数据操作等。

(1) 设置文本显示RAM 地址子程序

```

CUL EQU 32H ; 列坐标地址(以字节为单位)
ROW EQU 34H ; 行坐标地址(以字符行为单位)
WR_ADD: MOV A,ROW ; 取行坐标地址
      MOV B,#28H ; 设置显示域宽度(依实际而定)
      MUL AB ; 计算光标指针地址
      ADD A,CUL ; 加列坐标地址
      MOV CUL,A ; 光标低8位暂存CUL
      MOV A,B
      ADDC A,#00H
      MOV ROW,A ; 光标高8位暂存ROW

```



```

MOV      COM,#0AH      ; 设置光标指针
LCALL   PR1
MOV      DAT,CUL
LCALL   PR2
MOV      COM,#0BH
LCALL   PR1
MOV      DAT,ROW
LCALL   PR2
RET
    
```

(2) 西文字符组写入子程序

```

COUNT  EQU      35H      ; 写入字符个数
DPTR    为字符组首地址
WR_C:   PUSH     DPL      ; 暂存DPTR
        PUSH     DPH
        MOV      COM,#0CH ; 设置写数据指令代码
        LCALL   PR1
        POP      DPH      ; 取出DPTR
        POP      DPL
WR_C1:  CLR      A
        MOVC    A,@A+DPTR ; 取字符代码
        INC     DPTR      ; 修正DPTR
        PUSH     DPL      ; 暂存DPTR
        PUSH     DPH
        MOV      DAT,A    ; 写入显示数据
        LCALL   PR2
        POP      DPH      ; 取出DPTR
        POP      DPL
        DJNZ    COUNT,WR_C1 ; 循环
        RET
    
```

西文字符写入演示程序段

```

MOV      CUL,#00H      ; 设置列坐标为00
MOV      ROW,#00H     ; 设置行坐标为00
LCALL   WR_ADD        ; 调设置文本显示RAM地址子程序
MOV      DPTR,#CTAB   ; 设置字符组首地址
MOV      COUNT,#1AH   ; 写入字符个数为26
LCALL   WR_C          ; 调写入西文字符组子程序
SJMP    $
    
```

西文字符表

```

CTAB:DB 53H,75H,70H,70H,6FH,72H,74H,20H,54H,45H,4CH,20H
      DB 3AH,20H,30H,31H,30H,2DH,36H,32H,37H,38H,30H,33H,37H,39H
    
```

关光标,开闪烁程序(观察显示效果)

```

MOV      COM,#00H
LCALL   PR1
MOV      DAT,#38H
LCALL   PR2
    
```

```
SJMP      $
```

读数据示例程序段

```

MOV      CUL,#00H      ; 设置列坐标地址为00
MOV      ROW,#00H     ; 设置行坐标地址为00
LCALL   WR_ADD        ; 调文本显示RAM地址子程序
MOV      COM,#0DH     ; 设置读数据指令代码
LCALL   PR1
LCALL   PR3           ; "空读"

LOOPB:   LCALL        PR3      ; 读显示数据
MOV      A,#3AH       ; 判所读数据是":"否?
CJNE    A,DAT,LOOPB  ; 不是则再读下一个单元
MOV      CUL,#0CH     ; 是则将":"改写成"-"
MOV      ROW,#00H
LCALL   WR_ADD

```

单个西文字符写入程序段

```

MOV      COM,#0CH     ; 设置写数据指令代码
LCALL   PR1
MOV      DAT,#2DH    ; 写字符"-"代码
LCALL   PR2
SJMP    $

```

示例四 汉字字符写入

由于HD61830显示数据传输是从D0位开始的，所以在制作图形时要注意数据生成的顺序。汉字字符的写入本身就是规则图形的写入，而汉字位生成顺序是由D7位开始的，所以程序要有汉字字模结构的转换子程序。该程序使用的汉字库是由精电蓬远公司提供的汉字提取软件PICKHZB.EXE生成的。

写汉字字符子程序

```

CUL      EQU      32H      ; 列坐标地址(以字节为单位)
ROW      EQU      33H      ; 行坐标地址(以点行为单位)
CODE     EQU      36H      ; 汉字代码(单字节形式)
COUNT   EQU      35H      ; 计数器

WR_CC:   MOV      A,CODE
MOV      DPTR,#CCTAB
MOV      B,#20H
MUL     AB
ADD     A,DPL
PUSH    ACC
MOV     A,B
ADDC   A,DPH
PUSH    ACC
MOV     A,ROW      ; 取行坐标地址
MOV     B,#28H     ; 设置显示域宽度(依实际而定)
MUL     AB         ; 计算光标指针地址
ADD     A,CUL      ; 加列坐标地址
MOV     CUL,A     ; 地址低8位暂存CUL
MOV     A,B

```

	ADDC	A,#00H	
	MOV	ROW,A	; 地址高8位暂存ROW
	MOV	COUNT,#00H	; 计数器清零
WR_CC1:	MOV	COM,#0AH	; 设置光标地址
	LCALL	PR1	
	MOV	DAT,CUL	
	LCALL	PR2	
	MOV	COM,#0BH	
	LCALL	PR1	
	MOV	DAT,ROW	
	LCALL	PR2	
	MOV	COM,#0CH	; 写数据指令代码
	LCALL	PR1	
	POP	DPH	
	POP	DPL	; 设置汉字库首址
	MOV	A,COUNT	
	MOVC	A,@A+DPTR	; 取汉字左半部字模数据
	MOV	DAT,A	; 写入字模数据
	PUSH	DPL	
	PUSH	DPH	
	LCALL	CONVERT	; 调字模结构转换子程序
	LCALL	PR2	
	POP	DPH	
	POP	DPL	
	MOV	A,COUNT	
	ADD	A,#10H	; 修改间址参数
	MOVC	A,@A+DPTR	; 取汉字右半部字模数据
	MOV	DAT,A	; 写入字模数据
	PUSH	DPL	
	PUSH	DPH	
	LCALL	CONVERT	; 调字模结构转换子程序
	LCALL	PR2	
	MOV	A,CUL	; 修正光标指针地址
	ADD	A,#28H	; 加显示域宽度(依实际而定)
	MOV	CUL,A	
	MOV	A,ROW	
	ADDC	A,#00H	
	MOV	ROW,A	
	INC	COUNT	; 计数器加一
	MOV	A,COUNT	
	CJNE	A,#10H,WR_CC1	
	POP	ACC	
	POP	ACC	
	RET		
汉字字模结构转换子程序			
CONVERT:	MOV	A,COUNT	; 暂存COUNT计数器
	PUSH	ACC	
	MOV	COUNT,#08H	; 借用COUNT作为计数器

```

MOV      A,DAT      ; 取数据
MOV      DAT,#00H   ; DAT清零作为新数据
CONVER1: RRC        ; ACC.0→C
XCH      A,DAT      ; 新旧数据交换
RLC      A          ; C→ACC.0
XCH      A,DAT      ; 新旧数据交换
DJNZ     COUNT,CONVER1 ; 循环
POP      ACC        ; 恢复COUNT
MOV      COUNT,A
RET

```

汉字库（该汉字库是由精电蓬远公司提供的汉字提取软件PICKHZB.EXE生成的。）

```

CCTAB:   DB 000H,040H,037H,010H,083H,060H,02FH,008H;清=00H
          DB 013H,022H,0E3H,022H,023H,022H,022H,022H
          DB 040H,048H,0FCH,040H,0F8H,040H,0FEH,008H
          DB 0FCH,008H,0F8H,008H,0F8H,008H,028H,010H
          DB 004H,004H,008H,008H,018H,029H,04AH,008H;华=01H
          DB 009H,001H,0FFH,001H,001H,001H,001H,001H
          DB 040H,048H,058H,060H,0C0H,040H,044H,044H
          DB 03CH,000H,0FEH,000H,000H,000H,000H,000H
          DB 004H,0FFH,005H,041H,033H,014H,001H,0F6H;蓬=02H
          DB 01BH,010H,013H,010H,017H,010H,028H,047H
          DB 044H,0FEH,040H,0F8H,010H,0E0H,018H,046H
          DB 0F8H,040H,0F8H,040H,0FCH,040H,046H,0FCH
          DB 000H,043H,030H,010H,000H,007H,0F1H,011H;远=03H
          DB 011H,011H,012H,012H,014H,028H,047H,000H
          DB 010H,0F8H,000H,000H,008H,0FCH,020H,020H
          DB 020H,024H,024H,01CH,000H,006H,0FCH,000H

```

汉字显示演示程序

```

MOV      CUL,#00H
MOV      ROW,#00H
MOV      CODE,#00H
LCALL   WR_CC
MOV      CUL,#02H
MOV      ROW,#08H
MOV      CODE,#01H
LCALL   WR_CC
MOV      CUL,#04H
MOV      ROW,#10H
MOV      CODE,#02H
LCALL   WR_CC
MOV      CUL,#06H
MOV      ROW,#18H
MOV      CODE,#03H
LCALL   WR_CC
SJMP    $

```

附录一：HD61830内藏字符集

Lower 4 bits	Higher 4 bits	0010	0011	0100	0101	0110	0111	1010	1011	1100	1101	1110	1111
xxxx0000		0	a	P	`	P		-	夕	三	α	ρ	
		!	1	A	Q	a	q	u	ア	チ	△	ä	q
xxxx0010		"	2	B	R	b	r	「	イ	ツ	×	β	θ
		#	3	C	S	c	s	」	ウ	テ	ε	ε	ω
xxxx0100		\$	4	D	T	d	t	、	工	ト	†	μ	α
		%	5	E	U	e	u	・	オ	ナ	1	ε	ü
xxxx0110		&	6	F	V	f	v	ヲ	カ	ニ	ヨ	ρ	Σ
		'	7	G	W	g	w	ア	キ	ヌ	ラ	g	π
xxxx1000		(8	H	X	h	x	イ	ウ	ネ	リ	フ	×
)	9	I	Y	i	y	ウ	ク	ル		"	y
xxxx1010		*	#	J	Z	j	z	エ	コ	ン	レ	j	〒
		+	;	K	L	k	l	（	オ	ウ	ヒ	ロ	*
xxxx1100		,	<	L	¥	1	1	†	シ	フ	ワ	φ	π
		-	=	M	I	m	i	ユ	ズ	△	△	ε	*
xxxx1110		.	>	N	^	n	+	ヨ	セ	ホ	°	π	
		/	?	0	2	0	+	ツ	ツ	マ	"	ö	
xxxx1111													

附录二：蓬远公司配套产品

- (1) 各种液晶显示器件背光所需的逆变器，有PYE-D32-50、PYE-D32-49、NINV-1(配EL背光)；CXA-L10A(配CCFL背光)。
- (2) 日本OPTREX、SHARP等公司产品所需控制器及对应控制板（卡），有SED1335控制板（共四种版本）
- (3) 各种液晶模块的底层演示板。
- (4) 负电压产生模块DC-DC变换器。

附录三：精电蓬远公司维修服务规范

随着LCM的应用越来越广泛，LCM的售后维修服务工作越来越重要。为了能长期为用户提供良好的售后服务，公司特制定有关LCM的维修范围及维修项目的收费标准。

1. LCM不属于设备，而是作为元器件提供给用户。因此购买LCM的客户不享受保修服务。
2. 无偿维修服务：由于产品质量造成LCM显示不正常，及外观缺陷，公司将提供免费维修，必要时可调换商品。
3. 有偿维修服务：凡由于人为造成LCM的外观及电路的损坏，以至造成LCM无法正常显示，客户应承担一定数额的维修成本费用。
4. 本公司自行设计、制做的各种板、卡的维修服务，具体办法参照LCM的规范。
5. 对于本公司销售的LCM及相关产品，客户在使用过程中一旦发现问题，请及时与本公司联系，切勿自行处理。

附1 LCM的常见故障（以下故障条目，由香港精电公司提供）

项目	故障现象
液晶片	缺行/列；共用极/图画极短路；字暗/行暗；液晶外观问题；电压不当；多字画；显示模糊；彩虹；气泡；黑点；针孔；液晶片损坏；偏振片擦花；玻璃碎/玻璃裂
线路板	线路板弯曲；线路板短路、断路；电镀孔断路；线路损坏；焊盘损坏
背光	背光不平均；扩散膜不平；背光板损坏；背光灯损坏；背光逆变器损坏
斑马条	斑马条扭曲；斑马条移位
热压导电膜	热压片损坏；热压片焊接不良
外框	外框损坏；外框松动
芯片	IC失效；IC管脚中有异物

修订日期：2001年7月


附录四：内藏 HD61830 控制驱动器图形液晶显示模块目录及其尺寸图

尺寸单位: mm


点数	型号	点尺寸 W×H	像素尺寸 W×H	可视区尺寸 W×H	模块尺寸 W×H	内藏 控制器	背光类型					
							LED 背光			EL	CCF	
							01	02	03			04
80×32	MGL(S)-8032B	0.43×0.43	0.48×0.48	46.0×18.0	84.0×44.0	HD61830		√			√	
84×64	MGL(S)-8464	0.50×0.54	0.55×0.59	51.0×40.0	90.0×60.0	HD61830						√
160×80	MGL(S)-16080	0.39×0.39	0.42×0.42	72.3×37.8	100.0×54.0	HD61830				√		
240×128	MGL(S)-240128	0.40×0.40	0.45×0.45	114.0×64.0	144.0×104.0	HD61830			√	√		
240×128	MGL(S)-240128-CCFL	0.40×0.40	0.45×0.45	114.0×64.0	153.0×104.0	HD61830						√
320×64	MGL(S)-32064	0.36×0.36	0.41×0.41	136.8×33.6	165.0×45.0	HD61830						√

MGL(S)-8032B

80 X 32 DOTS




16 PIN CONNECTION	1	2	3	4	5	6	7	8	9	10	11	12	13	14	15	16
	VSS	VDD	V0	RS	R/W	E	DB0	DB1	DB2	DB3	DB4	DB5	DB6	DB7	/CS	/RST




MGL(S)-8464

84 X 64 DOTS




PIN	1	2	3	4	5	6	7	8	9	10	11	12	13	14	15	16
	VSS	VDD	V0	RS	R/W	E	DB0	DB1	DB2	DB3	DB4	DB5	DB6	DB7	/CS	/RES




* HD61830 OR EQUIVALENT

MGL(S)-16080

160 X 80 DOTS




1	2	3	4	5	6	7	8	9	10	11	12	13	14	15	16	17	18	19	20
VSS	VDD	VO	RS	R/W	E	DB0	DB1	DB2	DB3	DB4	DB5	DB6	DB7	CS	RES	VEE	LED-	N.C.	LED+




* HD61830 OR EQUIVALENT

MGL(S)-240128

240 X 128 DOTS




1	2	3	4	5	6	7	8	9	10	11	12	13	14	15	16	17	18
VSS	VDD	V0	RS	R/W	E	DB0	DB1	DB2	DB3	DB4	DB5	DB6	DB7	/CS	/RST	LED+	LED-




MGL(S)-240128-CCFL

240X 128 DOTS




1	2	3	4	5	6	7	8	9	10	11	12	13	14	15	16	17	18
VSS	VDD	V0	RS	R/W	E	DB0	DB1	DB2	DB3	DB4	DB5	DB6	DB7	/CS	/RST	NC	NC

* HD61830 OR EQUIVALENT




MGL(S)-32064

320 X 64 DOTS



16 PIN CONNECTION	1	2	3	4	5	6	7	8	9	10	11	12	13	14	15	16
	VSS	VDD	V0	RS	R/W	E	DB0	DB1	DB2	DB3	DB4	DB5	DB6	DB7	/CS	/RST



* HD61830 OR EQUIVALENT

地址: 北京市海淀区清华大学科技园华业大厦 3 区 2 层

邮编: 100084

开户名: 北京精电蓬远显示技术有限公司

开户行: 北京建行清华园分理处

账号: 6510005092610025020

税号: 110108801145824

电话: 010-62780866

传真: 010-62771832

联系人: 王学恩 马迎春

模块维修部联系人: 王雨生

网址: <http://www.vp-display.com>

E-mail: support@vp-display.com